

Lecture 10

Pseudo Instructions, Branching, & Control Structures in MIPS

```
## calculate A*X^2+B*X+C
## Output format must be:
## "answer = 180"
#####
#                               #
#####

.text
.globl main
main:
    lw $t0,X
    lw $t1,A
    lw $t2,B
    lw $t3,C

    mul $t4,$t0,$t0 # t4 = X^2
    mul $t4,$t4,$t1 # t4 = A*X^2
    mul $t5,$t2,$t0 # t5 = B*X
    add $t4,$t4,$t5 # t4 = A*X^2+B*X
    add $t4,$t4,$t3 # t4 = A*X^2+B*X+C

    la $a0,ans      # system call to print
    li $v0,4 # out string
    syscall

    move $a0,$t4    # print result on terminal
    li $v0,1
    syscall

    la $a0,end1    # system call to print
    li $v0,4 # out a newline
    syscall

    li $v0,10
    syscall        # au revoir

#####
#                               #
#####

.data
X: .word 7
A: .word 3
B: .word 4
C: .word 5
ans: .asciiz "answer = "
end1: .asciiz "\n"
```

This code is available to you on Sparky,
in directory ~cse220/examples/spim

← Load values of x, A, B, and C into temp registers

← Register \$t4 holds final value

← Print the string "answer = "

← Print the result value

← Print the newline

← Exit the program

- *mul* is a pseudo instruction. Actual instruction is *mult*.
 - There is no actual corresponding machine language instruction.
 - The assembler will replace the *mul* instruction with one or more assembly instructions.
 - Why? For programming convenience.
 - Ex: In this program, we use `mul $t4, $t0, $t0` to calculate x^2 , and store it in `$t4`.
 - Assembler does the following:


```
mult $t0, $t0
mflo $t4
```
 - Since `$t0` contains x , `mult` instruction will calculate x^2 in registers Hi and Lo.
 - Instruction `mflo` will copy lower 32 bits of product to `$t4`.
 - Thus `$t4` will have the product.
 - This requires that the product is less than 32 bits in length.

- Best way to study how pseudo instructions are assembled is to use them in a program and run it using SPIM.
 - Try:
 - `move $t1, $t2`
 - `add $t1, $zero, $t2`

- Several branch instructions are actually pseudo-instructions [conditional].
 - `la` (load address) is also a pseudo instruction.
 - `lw $t0, value` is also assembled slightly differently. Address of variable *value* is 32 bits long and we can not fit that in one instruction.

Branching

- Branch instructions allow for the changing of the flow of the program. Without them, the program will continue in a straight line of execution.
- Two types of branch instructions
 - Conditional branches
 - Unconditional/Direct branches
- *Conditional branches*
 - Branch which is dependent on a comparison. This comparison can be between 2 values in registers or with zero, and can be a signed or unsigned comparison.
 - Ex:
 - `beq` # branch on equal
 - `bgt` # branch on greater than
 - `beqz` # branch on equal to zero
 - Many conditional branches are pseudo-instructions. They are assembled using various 'set' instructions.
 - Ex:
 - `bge $t4, $t2, label` # branch to label if \$t4 >= \$t2
 - Assembled as:
 - `slt $1, $t4, $t2` # set register \$1 to 1, if register \$t4 < register \$t2
 - `beq $1, $0, label` # branch to label if \$1 == 0
 - Register \$t4 is \$t4, register \$t2 is \$t2.
 - So if \$t4 < \$t2, then set \$1 to 1, or \$1 is 0 if \$t4 >= \$t2. This is what we want.
 - Next instruction branches if register \$1 is 0.
 - This requires that register \$1 is available to the assembler. (\$1 is same as \$at).
- *Unconditional branch*
 - Branch which always occurs.
 - Ex:
 - `b` # branch to label
- *Jump instructions*
 - Several types of jump instructions
 - `jr $ra` # jump to address in register
 - Specifically, this instruction is how you return from a function (more on functions later).
 - Can replace \$ra with any register.

- `jal label` # jump and link
 - This instruction is how you jump to a function.
 - Saves the return address in the `$ra` register ($\$ra = PC + 4$), and then jumps to the label.
 - `jalr Rsrc` #jump and link register
 - This instruction jumps to the address stored in any register specified by `Rsrc` and places the return address in `$ra` ($\$ra = PC + 4$).
- Branch and jump instructions and their operand formats
 - Cross symbols represent pseudo-instructions.

Description	Opcode	Operands
Branch instruction	<code>b†</code>	label
Branch coprocessor z true	<code>bcz†</code>	label
Branch coprocessor z false	<code>bczf</code>	label
Branch on equal	<code>beq</code>	<code>Rsrc1, Src2, label</code>
Branch on equal zero	<code>beqz†</code>	<code>Rsrc, label</code>
Branch on greater than equal	<code>bge†</code>	<code>Rsrc1, Src2, label</code>
Branch on GTE unsigned	<code>bgeu†</code>	<code>Rsrc1, Src2, label</code>
Branch on greater than equal zero	<code>bgez</code>	<code>Rsrc, label</code>
Branch on greater than equal zero and link	<code>bgezal</code>	<code>Rsrc, label</code>
Branch on greater than	<code>bgt†</code>	<code>Rsrc1, Src2, label</code>
Branch on greater than unsigned	<code>bgtu†</code>	<code>Rsrc1, Src2, label</code>
Branch on greater than zero	<code>bgtz</code>	<code>Rsrc, label</code>
Branch on less than equal	<code>ble†</code>	<code>Rsrc1, Src2, label</code>
Branch on LTE unsigned	<code>bleu†</code>	<code>Rsrc1, Src2, label</code>
Branch on less than equal zero	<code>blez</code>	<code>Rsrc, label</code>
Branch on greater than equal zero and link	<code>bgezal</code>	<code>Rsrc, label</code>
Branch on less than and link	<code>bltzal</code>	<code>Rsrc, label</code>
Branch on less than	<code>blt†</code>	<code>Rsrc1, Src2, label</code>
Branch on less than unsigned	<code>bltu†</code>	<code>Rsrc1, Src2, label</code>
Branch on less than zero	<code>bltz</code>	<code>Rsrc, label</code>
Branch on not equal	<code>bne</code>	<code>Rsrc1, Src2, label</code>
Branch on not equal zero	<code>bnez†</code>	<code>Rsrc, label</code>
Jump	<code>j</code>	label
Jump and link	<code>jal</code>	label
Jump and link register	<code>jalr</code>	<code>Rsrc</code>
Jump register	<code>jr</code>	<code>Rsrc</code>

Control Structure

- Any reasonable programming language must have an *if* statement, and some type of *loop*.
- Otherwise programs would be straight lines of code.
- Need conditional branch instructions and a direct or unconditional branch instruction.
- Ex:
 - `beq $t0, $t1, label` # if $\$t0 == \$t1$ then jump to the address of label
 - `beqz $t0, label` # if $\$t0 == 0$ then jump to the address of label
 - `bne $t0, $t1, label` # if $\$t0 != \$t1$ then jump to the address of label
 - `j label` # jump to the address of label (direct branch)

- Code Structure for *if* and *while* statements
 - If <condition> then <then_part>, else <else_part>

Code for testing or evaluating condition <ul style="list-style-type: none"> - Loading/calculating values - Loading/calculating addresses
--

Branch statement to label L1 (Ex: <code>bne \$t0, \$t1, L1</code>)
--

Code for <else_part>
<code>j Loop_Done</code>

L1:	Code for <then_part>
------------	----------------------


Loop_Done:	Code for <loop_done>
-------------------	----------------------

Implementation of an 'if'

◦ **Compile by hand**

C `if (i == j) f=g+h;`
`else f=g-h;`

Mapping `f: $s0, g: $s1,`
`h: $s2, i: $s3, j: $s4`



M `beq $s3, $s4, True # branch i==j`
I `sub $s0, $s1, $s2 # f=g-h (false)`
`j Exit # go to Exit`
P `True: add $s0, $s1, $s2 # f=g+h (true)`
S `Exit:`

Implementation of a 'do-while'

```
C      Loop:   g = g + A[i];  
        i = i + j;  
        if (i != h) goto Loop;
```

(g,h,i,j:\$s1,\$s2,\$s3,\$s4 : base of A[]:\$s5)

```
M      Loop:  add $t1,$s3,$s3  # $t1 = 2*i  
I          add $t1,$t1,$t1  # $t1 = 4*i  
P          add $t1,$t1,$s5  # $t1 = addr A  
S          lw  $t1,0($t1)   # $t1 = A[i]  
          add $s1,$s1,$t1  # g = g + A[i]  
          add $s3,$s3,$s4  # i = i + j  
          bne $s3,$s2,Loop # goto Loop  
                    # if i != h
```

Implementation of a 'while'

```
C      while (save[i]==k)  
        i = i + j;
```

(i,j,k: \$s3,\$s4,\$s5: base of save[]:\$s6)

```
M      Loop:  add $t1,$s3,$s3  # $t1 = 2*i  
I          add $t1,$t1,$t1  # $t1 = 4*i  
P          add $t1,$t1,$s6  # $t1 = Addr  
S          lw  $t1,0($t1)   # $t1 = save[i]  
          bne $t1,$s5,Exit # goto Exit  
                    # if save[i] != k  
          add $s3,$s3,$s4  # i = i + j  
          j    Loop        # goto Loop
```

Exit:

